// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: contracts/Swap0x.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IMulticall {
    function multicall(bytes[] calldata data)
        external
        payable
        returns (bytes[] memory results);
}

abstract contract Multicall is IMulticall {
    function _getRevertMsg(bytes memory _returnData)
        internal
        pure
        returns (string memory)
    {
        // If the _res length is less than 68, then the transaction failed silently (without a revert message)
        if (_returnData.length < 68) return "Transaction reverted silently";

        assembly {
            // Slice the sighash.
            _returnData := add(_returnData, 0x04)
        }
        return abi.decode(_returnData, (string)); // All that remains is the revert string
    }

    function multicall(bytes[] calldata data)
        public
        payable
        override
        returns (bytes[] memory results)
    {
        results = new bytes[](data.length);
        for (uint256 i = 0; i < data.length; i++) {
            (bool success, bytes memory result) = address(this).delegatecall(
                data[i]
            );

            require(success, _getRevertMsg(result));
            results[i] = result;
        }
    }
}

contract Swap0x is Multicall {
    receive() external payable {}

    function swap(
        address sellToken,
        address buyToken,
        uint256 sellAmount,
        address allowanceTarget,
        address payable swapTarget,
        bytes calldata swapData
    ) public payable {
        if (sellToken != address(0)) {
            IERC20(sellToken).transferFrom(
                msg.sender,
                address(this),
                sellAmount
            );
        } else {
            require(msg.value >= sellAmount, "Swap0x: ETH value invalid");
        }

        if (allowanceTarget != address(0) && sellToken != address(0)) {
            require(
                IERC20(sellToken).approve(allowanceTarget, type(uint256).max),
                "Swap0x: allowance failed"
            );
        }

        uint256 buyTokenBalanceBefore = 0;
        if (buyToken != address(0)) {
            buyTokenBalanceBefore = IERC20(buyToken).balanceOf(address(this));
        } else {
            buyTokenBalanceBefore = address(this).balance;
        }

        if (sellToken != address(0)) {
            (bool success, bytes memory retdata) = swapTarget.call(swapData);
            require(success, _getRevertMsg(retdata));
        } else {
            (bool success, bytes memory retdata) = swapTarget.call{
                value: sellAmount
            }(swapData);
            require(success, _getRevertMsg(retdata));
        }

        if (buyToken != address(0)) {
            uint256 buyTokenBalanceAfter = IERC20(buyToken).balanceOf(
                address(this)
            );
            uint256 boughtAmount = buyTokenBalanceAfter - buyTokenBalanceBefore;
            IERC20(buyToken).transfer(msg.sender, boughtAmount);
        } else {
            uint256 buyTokenBalanceAfter = address(this).balance;
            uint256 boughtAmount = buyTokenBalanceAfter - buyTokenBalanceBefore;
            payable(msg.sender).transfer(boughtAmount);
        }
    }
}